

LAB 1: Linear Mixed Models

Analyze the fecal fat data (fecfat) using SAS:

- 1) Calculate estimates of the mean value for each of the patients.
- 2) Fit a model using SAS Proc GLM with outcome of fecal fat and the predictors being pill type and subject.
- 3) Fit the same model using SAS Proc MIXED (declare subject to be a random factor).
- 4) How do the analyses compare with respect to both results and fitting methods?
- 5) Obtain the predicted values for each of the patients using both GLM and MIXED. (Use an ESTIMATE statement and specify the intercept as well). How do these compare to each other and the values calculated in 1)? What is the explanation?
- 6) Now add sex to the model and analyze using GLM and MIXED. How do they compare?

```

proc sort data=work.fecfat;
by subject;
run;
proc means;
var fecfat;
by subject;
run;
proc glm;
class pilltype subject;
model fecfat=pilltype subject;
estimate 'subj 1' intercept 1 subject 1 0 0 0 0;
estimate 'subj 2' intercept 1 subject 0 1 0 0 0;
estimate 'subj 3' intercept 1 subject 0 0 1 0 0;
estimate 'subj 4' intercept 1 subject 0 0 0 1 0;
run;
proc mixed;
class pilltype subject;
model fecfat=pilltype;
random subject;
estimate 'subj 1' intercept 1 | subject 1 0 0 0 0;
estimate 'subj 2' intercept 1 | subject 0 1 0 0 0;
estimate 'subj 3' intercept 1 | subject 0 0 1 0 0;
estimate 'subj 4' intercept 1 | subject 0 0 0 1 0;
run;
proc glm;
class pilltype subject female;
model fecfat=pilltype subject female;
run;
proc mixed;
class pilltype subject female;
model fecfat=pilltype female;
random subject;
run;
proc mixed;
class pilltype subject female;
model fecfat=pilltype female/ddfm=kenrog;
random subject;
run;
proc glm;
class pilltype subject female;
model fecfat=pilltype female subject(female);
test h=female e=subject(female);
run;

```

```
library(nlme)
fecfat<-read.table("fecfat.txt", header=TRUE)
fecfat<-fecfat[order(fecfat[,"subject"]),]
lmeFit<-lme(fecfat ~ female + pilltype, random=~1 | subject,
data=fecfat)
summary(lmeFit)
#add intercept to random effects to get predicted values
fixef(lmeFit)[1]+ranef(lmeFit)
```

Analyze the fecal fat zero data (fecfatzero) using SAS. This is a dataset artificially created to have little subject to subject variation:

- 7) Rerun the GLM and MIXED analyses. How do the predicted values compare?
- 8) What does the log for MIXED indicate?
- 9) Run the MIXED analysis without the random statement.
- 10) Run the MIXED analysis using the DDFM=KENROG option on the model statement.
- 11) How do the tests for pilltypes compare across the analyses? What is the explanation?

```

proc sort data=work.fecfatzero;
by subject;
run;
proc means;
var fecfat;
by subject;
run;
proc glm;
class pilltype subject;
model fecfat=pilltype subject;
estimate 'subj 1' intercept 1 subject 1 0 0 0 0;
estimate 'subj 2' intercept 1 subject 0 1 0 0 0;
estimate 'subj 3' intercept 1 subject 0 0 1 0 0;
estimate 'subj 4' intercept 1 subject 0 0 0 1 0;
run;
proc mixed;
class pilltype subject;
model fecfat=pilltype;
random subject;
estimate 'subj 1' intercept 1 | subject 1 0 0 0 0;
estimate 'subj 2' intercept 1 | subject 0 1 0 0 0;
estimate 'subj 3' intercept 1 | subject 0 0 1 0 0;
estimate 'subj 4' intercept 1 | subject 0 0 0 1 0;
run;
proc mixed;
class pilltype subject;
model fecfat=pilltype;
run;
proc mixed;
class pilltype subject;
model fecfat=pilltype/ddfm=kenrog;
random subject;
run;

```

```
library(nlme)
fecfat0<-read.table("fecfatzero.txt", header=TRUE)
fecfat0<-fecfat[order(fecfat0["subject"]),]
lmeFit0<-lme(fecfat ~ female + pilltype, random=~1 | subject,
data=fecfat0)
summary(lmeFit0)
fixef(lmeFit0)[1]+ranef(lmeFit0)
```